
The basic idea of natural deduction in Martin-Löf’s style is to let each logical connective unfold into its

meta-level representation in terms of hypothetical and parametric judgments. The proof really takes place
in the meta-language, using the properties of these judgments. Logical connectives are internalizations of
pre-existing meta-level judgments.

The question arises: how concise can we be? What is the minimal number of connectives needed to inter-
nalize all the essential features of the meta-language (hyp. and para. judgments)?

The answer, perhaps surprisingly, is one connective, together with function types in the term language. The

steps of the construction are:

1. Parametric judgments are converted to meta-functions.

2. Hypothetical judgments are converted to meta-equality.

3. Meta-functions are internalized by object functions.

4. Meta-equality is internalized by object equality.

First, we establish the syntactic categories of types, terms, and proofs.

tp : type.

tm : tp -> type.

o : tp.

pf : tm o -> type.

We define some preliminary concepts surrounding meta-equality (Leibniz equality).

eqv = [P] [Q] (pf P -> pf Q) & (pf Q -> pf P).

eq = [S] [T] {p:tm A -> tm o} eqv (p S) (p T).

feq = [F] [G] {p:(tm A -> tm B) -> tm o} eqv (p F) (p G).

Now we introduce axioms for each step of the construction.

1. Parametric judgments are converted to meta-functions.

abs : ({x} eq (F x) (G x)) -> feq F G.

The converse of abs is already a consequence of the general properties of hypothetico-parametric judgments.

2. Hypothetical judgments are converted to meta-equality.

oext : eqv P Q -> eq P Q.

Again, the converse follows by general properties of hypothetico-parametric judgments.

3. Meta-functions are internalized by object functions.

--> : tp -> tp -> tp.

lam : (tm A -> tm B) -> tm (A --> B).

app : tm (A --> B) -> (tm A -> tm B).

beta : feq (app (lam F)) F.

eta : eq (lam (app T)) T.

4. Meta-equality is internalized by object equality.

== : tm A -> tm A -> tm o.

in : eq S T -> pf (S == T).

out : pf (S == T) -> eq S T.

The definition of all the usual connectives of intuitionistic higher-order logic in terms of == is left as an
exercise for the reader. I conjecture that the provable closed formulae of this system are exactly those of
HOL Light (without set types, infinity, or choice).


